// File: src/Pledge.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.13;
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/utils/ReentrancyGuard.sol";


/**
 *   __ _ _ __   _____  ____      ____ _| |_ ___| |__   ___| |_   _| |__  
 *  / _` | '_ \ / _ \ \/ /\ \ /\ / / _` | __/ __| '_ \ / __| | | | | '_ \ 
 * | (_| | |_) |  __/>  <  \ V  V / (_| | || (__| | | | (__| | |_| | |_) |
 *  \__,_| .__/ \___/_/\_\  \_/\_/ \__,_|\__\___|_| |_|\___|_|\__,_|_.__/ 
 *       |_|                                                              
 *
 * @title Ownable USDT/USDC Pledge Contract
 *
 * @dev This contract is specifically built for Apex Watch Club Pledge
 */
contract Pledge is Ownable, ReentrancyGuard {
  // #############################
  // #         VARIABLES         #
  // #############################
  
  /**
   * @dev This mapping is used to determine how many a certain address pledged
   */
  mapping(address => uint256) _pledged;

	bool _isFrozen = true;
	uint256 _pledgePrice;
	uint256 _totalPledgedCount;
	uint256 _totalSupply;
	IERC20 _tokenUsdc;
	IERC20 _tokenUsdt;

  // #############################
  // #          EVENTS           #
  // #############################

	event PledgeUsdt(address _from, address _to, uint256 _amount, uint256 _timestamp);
	event PledgeUsdc(address _from, address _to, uint256 _amount, uint256 _timestamp);
	event WithdrawUsdt(address _to, uint256 _amount, uint256 _timestamp);
	event WithdrawUsdc(address _to, uint256 _amount, uint256 _timestamp);
	event Freeze(bool _status, uint256 _timestamp);
	event Unfreeze(bool _status, uint256 _timestamp);
	event PriceSet(uint256 _old, uint256 _new, uint256 _timestamp);
	event PledgedCountSet(uint256 _old, uint256 _new, uint256 _timestamp);
	event SupplySet(uint256 _old, uint256 _new, uint256 _timestamp);

	constructor(address _usdt, address _usdc, uint256 _supply, uint256 _price) Ownable(address(msg.sender)) ReentrancyGuard() {
		_tokenUsdt = IERC20(_usdt);
		_tokenUsdc = IERC20(_usdc);
		_pledgePrice = _price;
		_totalSupply = _supply; 
	}

  // #############################
  // #      PLEDGE FUNCTIONS     #
  // #############################

	function pledgeUsdt(uint256 _amount) public nonReentrant returns (bool) {
		// only admin can change state when frozen
		require(!_isFrozen || msg.sender == owner(), "All state changing transactions currently frozen.");
		// must be a multiple of _pledgePrice
		require(_amount % _pledgePrice == 0, 'Pledge amount must be a multiple of current price.');
		// must not exceed set supply
		uint256 _count = _amount / _pledgePrice;
		require(_totalPledgedCount + _count < _totalSupply, "Not enough supply. Consider reducing your pledge count.");
		// must have enough allowance
		require(_tokenUsdt.allowance(msg.sender, address(this)) >= _amount, 'USDT allowance not enough.');
		// must have enough USDT
		require(_amount <= _tokenUsdt.balanceOf(msg.sender), "Insufficient USDT balance.");

		_pledged[msg.sender] = _pledged[msg.sender] + _amount;
		_totalPledgedCount = _totalPledgedCount + _count;

		emit PledgeUsdt(msg.sender, address(this), _amount, block.timestamp);
		return  _tokenUsdt.transferFrom(msg.sender, address(this), _amount);
	}


	function pledgeUsdc(uint256 _amount) public nonReentrant returns (bool) {
		// only admin can change state when frozen
		require(!_isFrozen || msg.sender == owner(), "All state changing transactions currently frozen.");
		// must be a multiple of _pledgePrice
		require(_amount % _pledgePrice == 0, 'Pledge amount must be a multiple of current price.');
		// must not exceed set supply
		uint256 _count = _amount / _pledgePrice;
		require(_totalPledgedCount + _count < _totalSupply, "Not enough supply. Consider reducing your pledge count.");
		// must have enough allowance
		require(_tokenUsdc.allowance(msg.sender, address(this)) >= _amount, "USDC allowance not enough.");
		// must have enough USDC
		require(_amount <= _tokenUsdc.balanceOf(msg.sender), "Insufficient USDC balance.");

		_pledged[msg.sender] = _pledged[msg.sender] + _amount;
		_totalPledgedCount = _totalPledgedCount + _count;

		emit PledgeUsdc(msg.sender, address(this), _amount, block.timestamp);
		return _tokenUsdc.transferFrom(msg.sender, address(this), _amount);
	}

  // #############################
  // #       VIEW FUNCTIONS      #
  // #############################

	function getPledged(address _address) public view returns (uint256) {
		return _pledged[_address];
	}


	function getTotalPledgedCount() public view returns (uint256) {
		return _totalPledgedCount;
	}
	

	function getTotalSupply() public view returns (uint256) {
		return _totalSupply;
	}


	function getPrice() public view returns (uint256) {
		return _pledgePrice;
	}


  function getIsFrozen() public view returns (bool) {
    return _isFrozen;
  }


  // #############################
  // #      OWNER FUNCTIONS      #
  // #############################

	function setPrice(uint256 _newPrice) public onlyOwner returns (uint256) {
		// only admin can change state when frozen
		require(!_isFrozen || msg.sender == owner(), "All state changing transactions currently frozen.");
		require(_newPrice > 0, "New price must be greater than 0.");

		emit PriceSet(_pledgePrice, _newPrice, block.timestamp);

		_pledgePrice = _newPrice;
		return _pledgePrice;
	}

	function setTotalPledgedCount(uint256 _newTotalPledgedCount) public onlyOwner returns (uint256) {
		// only admin can change state when frozen
		require(!_isFrozen || msg.sender == owner(), "All state changing transactions currently frozen.");

		emit PledgedCountSet(_totalPledgedCount, _newTotalPledgedCount, block.timestamp);

		_totalPledgedCount = _newTotalPledgedCount;
		return _totalPledgedCount;
	}

	function setTotalSupply(uint256 _newSupply) public onlyOwner returns (uint256) {
		// only admin can change state when frozen
		require(!_isFrozen || msg.sender == owner(), "All state changing transactions currently frozen.");

		emit SupplySet(_totalSupply, _newSupply, block.timestamp);

		_totalSupply = _newSupply;
		return _totalSupply;
	}


  /**
   * @dev Prevent all non-admin transactions in case of emergency
   */
	function freeze() public onlyOwner {
		require(!_isFrozen, "Already frozen.");
		_isFrozen = true;
		emit Freeze(_isFrozen, block.timestamp);
	}
	

  /**
   * @dev Toggle opening / closing of pledging
   */
	function unfreeze() public onlyOwner {
		require(_isFrozen, "Not frozen.");
		_isFrozen = false;
		emit Unfreeze(_isFrozen, block.timestamp);
	}

	
	function withdrawUsdt() public onlyOwner nonReentrant returns (bool) {
		require(msg.sender == owner(), 'Only admin can withdraw USDT.');

		uint256 _amount = _tokenUsdt.balanceOf(address(this));

		emit WithdrawUsdt(msg.sender, _amount, block.timestamp);

		return _tokenUsdt.transfer(msg.sender, _amount);
	}


	function withdrawUsdc() public onlyOwner nonReentrant returns (bool) {
		require(msg.sender == owner(), 'Only admin can withdraw USDC.');

		uint256 _amount = _tokenUsdc.balanceOf(address(this));

		emit WithdrawUsdc(msg.sender, _amount, block.timestamp);

		return _tokenUsdc.transfer(msg.sender, _amount);
	}
}

// File: lib/openzeppelin-contracts/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (access/Ownable.sol)

pragma solidity ^0.8.20;

import {Context} from "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * The initial owner is set to the address provided by the deployer. This can
 * later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    /**
     * @dev The caller account is not authorized to perform an operation.
     */
    error OwnableUnauthorizedAccount(address account);

    /**
     * @dev The owner is not a valid owner account. (eg. `address(0)`)
     */
    error OwnableInvalidOwner(address owner);

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
     */
    constructor(address initialOwner) {
        if (initialOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(initialOwner);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        if (owner() != _msgSender()) {
            revert OwnableUnauthorizedAccount(_msgSender());
        }
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.20;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the value of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the value of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves a `value` amount of tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 value) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets a `value` amount of tokens as the allowance of `spender` over the
     * caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 value) external returns (bool);

    /**
     * @dev Moves a `value` amount of tokens from `from` to `to` using the
     * allowance mechanism. `value` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 value) external returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/utils/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/ReentrancyGuard.sol)

pragma solidity ^0.8.20;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant NOT_ENTERED = 1;
    uint256 private constant ENTERED = 2;

    uint256 private _status;

    /**
     * @dev Unauthorized reentrant call.
     */
    error ReentrancyGuardReentrantCall();

    constructor() {
        _status = NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be NOT_ENTERED
        if (_status == ENTERED) {
            revert ReentrancyGuardReentrantCall();
        }

        // Any calls to nonReentrant after this point will fail
        _status = ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == ENTERED;
    }
}

// File: lib/openzeppelin-contracts/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/Context.sol)

pragma solidity ^0.8.20;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}